



Temperaturni senzor LM35

Za ovu vežbu će se koristiti analogni senzor temperature LM35. Senzor na izlazu daje napon koji je linearano proporcionalan temperaturi, a za LM35 to je $10 \text{ mV}/^{\circ}\text{C}$. Za očitane vrednosti od npr. 220 mV , temperatura je 22.0°C . Samim tim se zaključuje da je dovoljno da se izlazna vrednost koja je očitana u mV podeli sa 10.



Slika 1: Pinovi senzora LM35



Slika 2: Povezivanje senzora na protoploči

Programski kod

U *datasheet*-u senzora nalazi se da je opseg merenja temperature od -55 °C do 150 °C, i zbog toga je potrebno podesiti da referentni napon za AD konvertor bude na 2.048 V.

```
Device = 18F25K22
Xtal 32
Dim T As Float
' =====
START:
' =====
ANSELA=%00000001      'RA0 pin selektovan kao analogni
TRISA=%00000001      'RA0 postavljen kao ulazni pin
VREFCON0=%11100000   'VREFCON0=2.048 V
ADCON0=%00000001     'ADC on
ADCON1=%00001000     'FVR buf za Vref+, GND za Vref-
ADCON2=%10111110     'Desno ravnanje, Fosc/64
' =====
MAIN:
' =====
ADCON0.1=1            'Pokrećemo konverziju
While ADCON0.1=1
Wend
T=(ADRESH * 256 + ADRESL)
T=T*2                 'jer je Vref 2*1024mV
T=T / 10              '10mV/oC
HserOut ["T=",Dec1 T, $B0, "C", 13,10]
'$B0 je simbol za stepen
DelayMS 2000
GoTo MAIN
End
```

Naredba **Symbol**

Kada je u pitanju komplikovaniji kod, desi se da je potrebno da se promeni neki pin, tada mora da se menja svaka linija koda koja koristi taj pin. Za takve potrebe uvodi se naredba **Symbol**, određenoj promenljivoj se dodeljuje naziv. Svaki sledeći put kada se koristi promenljiva upotrebljava se naziv koji je dodeljen, a samim tim kada želimo da promenimo pin ili vrednost u celom kodu, dovoljno je da se promeni samo u delu gde je deklarisan **Symbol**.


Primer:

```
...  
Symbol Crvena=PORTA.1  
Symbol Zelena=PORTA.2  
Symbol Plava=PORTA.3  
Symbol T1=25  
High Crvena  
Low Plava  
Low Zelena  
...
```

Naredba **Select...Case...EndSelect**

Ova naredba se koristi u slučaju kada je potrebno da na osnovu jednog izraza izvršimo jedan od ponuđenih uslova. Ovakva naredba se smatra unapređenom metodom If...Then...Else, zbog toga što sadrži više uslova od kojih je potrebno izvršiti jedan. U jednom uslovu moguće je izvršiti naredbu i za više vrednosti, za vrednosti manje ili veće od zadatog broja, čak i za vrednosti koje idu od do nekog broja.

Uključivanje diode na osnovu izmerene temperature



Slika 3: Povezivanje senzora LM35 i LED dioda

U zavisnosti od izmerene temperature, potrebno je uključivati i isključivati određene LED diode. Za temperaturu do 25 °C uključivaće se zelena LED, za temperaturu između 25 i 27°C će se uključivati žuta LED, a za temperaturu veću od 27°C uključivaće se crvena LED.

```

Device = 18F25K22
Xtal 32
Dim T As Float
'=====
START:
'=====
ANSELA=%00000001      'RA0 pin selektovan kao analogni
TRISA=%00000001      'RA0 postavljen kao ulazni pin
VREFCON0=%11100000   'VREFCON0=2.048 V
ADCON0=%00000001    'ADC on
ADCON1=%00001000    'FVR buf za Vref+, GND za Vref-
ADCON2=%10111110    'Desno ravnanje, Fosc/64
'=====
MAIN:
'=====
ADCON0.1=1
While ADCON0.1=1
Wend
T=(ADRESH * 256 + ADRESL)
T=T*2                  'jer je Vref 2*1024mV
T=T / 10              '10mV/oC
HserOut ["T=",Dec1 T, $B0, "C", 13,10]
'$B0 je simbol za stepen

Select T
    Case <25
        PORTA=%00001000      'Zelena LED
    Case 25 To 27
        PORTA=%00000100      'Žuta LED
    Case >27
        PORTA=%00000010      'Crvena LED
EndSelect

DelayMS 2000
GoTo Start

```